


구문분석

목차

- 알파벳, 스트링, 언어, 문법
- 문맥자유문법(context free grammar, cfg)
- BNF 표기법
- 유도(derivation)
- 유도 연습
- Leftmost/rightmost derivation
- 유도의 트리 표현
- Parsing
 - Top-down vs. bottom-up
- parse tree, AST
- 문법
 - Ambiguous grammar
 - Precedence, associativity
 - ◆ C 언어 연산자 우선 순위, 결합 규칙
 - Left recursion
 - Left factoring
- Recursive descent parsing
- Shift-reduce parsing



알파벳은 모두 다 String 이 된다
 이미 있는 String을 모아놓은 Language가 됨
 Language를 명정한 규칙에 맞게 나열하면
 Grammar가 된다

Alphabet, String, Language, Grammar

- 알파벳(alphabet)은 심볼(symbol)을 원소로 갖는, 공집합이 아닌, 유한집합이다. 스트링(string)은 알파벳 내 심볼(들)의 유한 나열(possibly empty)이다. 스트링 w 의 길이(length)는 w 내 심볼 개수이며 $|w|$ 로 표기한다. 어떤 심볼도 갖지 않는 스트링을 empty string이라 하며 λ 혹은 ϵ 으로 표기한다($|\lambda| = |\epsilon| = 0$). 언어(language)는 스트링의 집합이다. 언어에 속하는 스트링을 문장(sentence)이라 한다.

예) 심볼 a, b 를 원소로 갖는 알파벳 $\Sigma = \{a, b\}$

예) 알파벳 $\Sigma = \{a, b\}$ 로부터 얻을 수 있는 길이 1 이상 스트링들

$a, b, aa, ab, bb, aaa, bbb, aab, baa, aaaaaa, \dots$

예) 알파벳 $\Sigma = \{a, b\}$ 로부터 얻을 수 있는 언어

$L = \{ab, aabb, aaabbb, \dots\} = \{a^n b^n \mid n \geq 1\}$

언어 생성 수단으로서의 문법

언어 $L = \{ab, aabb, aaabbb, \dots\}$ 을 표현한 문법 G 의 예

$G = (V, T, S, P)$

- Non-terminal 심볼 집합 $V = \{S\}$
- Terminal 심볼 집합 $T = \{a, b\}$
- 시작 심볼 S
- 생성규칙(production rule) 집합 $P = \{S \rightarrow aSb, S \rightarrow ab\}$

$G = (V, T, S, P)$

문법 G 의 시작 심볼 S 에서 시작하여
 생성규칙들의 적용을 통해 언어 L 내 문장
 $ab, aabb, aaabbb$ 들이 생성(generate)되는
 유도(derivation) 과정들의 예

유도 과정을
 계속하면 44과 같은

$S \rightarrow ab$

$S \Rightarrow ab$

aSb 에 하나의 생성규칙 $S \rightarrow ab$ 가
 적용되어 aSb 가 $aabb$ 로 변경될 때
 aSb 로부터 $aabb$ 가
 유도(derivation)되었다고 하고
 $aSb \Rightarrow aabb$ 로 표기

$S \rightarrow aSb \quad S \rightarrow ab$

$S \Rightarrow aSb \Rightarrow aabb$

$S \rightarrow aSb \quad S \rightarrow ab$

$S \Rightarrow aSb \Rightarrow aaSbb \Rightarrow aaabbb$

문맥자유문법

문법 $G = (V, T, S, P)$ 는 다음과 같이 정의된다

- V 는 **variable(non-terminal symbol, 논터미널 심볼, 변수)**의 유한 집합
- T 는 **terminal symbol(터미널 심볼)**의 유한 집합
- S 는 **start variable(start symbol, 시작 심볼)**로 불리는 special symbol 이며 $S \in V$
- P 는 **production rule(생성 규칙)**의 유한 집합으로 각 production은 $\alpha \rightarrow \beta$ 형태이며 α, β 는 V, T 내 심볼로 이루어진 길이 0 이상 스트링이지만, α 는 V 내 심볼을 하나 이상 포함하고 있어야 함. 다른 언급이 없으면 V 와 T 는 non-empty이며 disjoint

문맥자유문법(context-free grammar, cfg)

- $G = (V, T, S, P)$ 의 모든 생성규칙들이 $A \rightarrow \alpha$ 의 형식이면 문법 G 는 context-free라고 한다. (A 는 V 의 원소이며 α 는 V, T 내 심볼로 이루어진 길이 0 이상 스트링)

$G = (V, T, S, P)$

$V = \{E, T, F\}$

$T = \{+, -, *, /, (,), id\}$

$S = E$

$P = \{E \rightarrow E + T, E \rightarrow E - T, E \rightarrow T, T \rightarrow T * F, T \rightarrow T / F, T \rightarrow F, F \rightarrow (E), F \rightarrow id\}$

- Nonterminal symbol(논터미널 심볼) $\rightarrow E, T, F$
- Terminal symbol(터미널 심볼) $\rightarrow +, -, *, /, (,), id$
- Start symbol(시작 심볼) $\rightarrow E$
- 생성규칙 개수 $\rightarrow 8$ 개

right-hand side(RHS)
Left-hand side(LHS)

$F \rightarrow (E)$
 $F \rightarrow id$

이렇게 축약 가능

$E \rightarrow E + T \mid E - T \mid T$
 $T \rightarrow T * F \mid T / F \mid F$
 $F \rightarrow (E) \mid id$

- 문법 G 를 위와 같이 생성규칙들만 나열하여 표현 가능
- 나열된 첫 생성규칙의 LHS 심볼이 시작 심볼이고, LHS가 동일한 규칙들은 기호 \mid 를 이용하여 축약 표기한 것으로 가정하면 위 생성규칙 나열로부터 문법 G 복원 가능

BNF 표기법

BNF(Backus-Naur Form) 표기법

- BNF 표기법은 John Backus와 Peter Naur에 의해 개발되었으며 프로그래밍언어의 문법 정의에 사용됨
- 메타 기호로 \langle , \rangle , $::=$, $|$ 를 사용하는데, 논터미널 심볼은 메타기호 \langle 와 \rangle 로 묶어 표기하고, 메타기호 $::=$ 은 **is defined as**의 의미로 \rightarrow 대신 사용하며, 메타기호 $|$ 은 **or**의 의미로 사용

BNF 표기법

```
E → E + T | E - T | T
T → T * F | T / F | F
F → ( E ) | id
```

```
<E> ::= <E> + <T> | <E> - <T> | <T>
<T> ::= <T> * <F> | <T> / <F> | <F>
<F> ::= ( <E> ) | id
```

Derivation

유도(derivation)

- 스트링 $w = u\alpha v$ 에 대해 **production rule** $\alpha \rightarrow \beta$ 가 적용가능(applicable)하다고 하며 $\alpha \rightarrow \beta$ 를 $w = u\alpha v$ 에 적용하면 w 내 α 를 β 로 대치(replace)하여 새로운 스트링 $z = u\beta v$ 를 얻게 되는데 이 과정을 $w \Rightarrow z$ 로 표기하고 w 가 z 를 **유도한다(derive)**라고 하거나 혹은 z 가 w 로부터 유도되었다고 한다.
Grammar의 production들을 임의 순서로 적용하여 새로운 스트링들이 유도될 수 있다

문법 G 에 의해 생성되는 언어

- 문법 G 에 의해 생성되는 언어는 G 의 시작 심볼 S 로부터 유도되는 모든 문장의 집합

아래 문법 G 로부터 $abba$ 의 유도(과정)

$$G = (\{S, A, B\}, \{a, b\}, S, \{S \rightarrow aA \mid bB \mid \lambda, A \rightarrow bS, B \rightarrow aS\})$$

$$\begin{aligned} S &\rightarrow aA \mid bB \mid \lambda \\ A &\rightarrow bS \\ B &\rightarrow aS \end{aligned}$$

$$S \Rightarrow aA \Rightarrow abS \Rightarrow abbB \Rightarrow abbaS \Rightarrow abba$$

Sentential forms Sentence

$S \Rightarrow aA$	(생성 규칙 $S \rightarrow aA$ 적용)
$\Rightarrow abS$	(생성 규칙 $A \rightarrow bS$ 적용)
$\Rightarrow abbB$	(생성 규칙 $S \rightarrow bB$ 적용)
$\Rightarrow abbaS$	(생성 규칙 $B \rightarrow aS$ 적용)
$\Rightarrow abba$	(생성 규칙 $S \rightarrow \lambda$ 적용)

- S 에 생성규칙 $S \rightarrow aA$ 가 적용되어 S 가 aA 로 변경될 때 S 로부터 aA 가 유도(derivation)되었다고 하고 $S \Rightarrow aA$ 로 표기
- aA 에 생성규칙 $A \rightarrow bS$ 가 적용되어 aA 가 abS 로 변경될 때 aA 로부터 abS 가 유도(derivation)되었다고 하고 $aA \Rightarrow abS$ 로 표기

유도 연습 1

해당 과정을 유도하라

문법 $G = (V, T, S, P)$
 $V = \{S, E, T, F, D\},$
 $T = \{+, *, (,), 0, 1, 2, 3, 4, 5, 6, 7, 8, 9\}$

문법 G의 생성규칙집합 P

$S \rightarrow E$
 $E \rightarrow E + T$
 $E \rightarrow T$
 $T \rightarrow T * F$
 $T \rightarrow F$
 $F \rightarrow (E)$
 $F \rightarrow D$
 $D \rightarrow 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9$

- 문법 G를 사용하여 $1 + 2$ 의 유도과정을 보이시오

S
 $\Rightarrow E$
 $\Rightarrow E + T$
 $\Rightarrow T + T$
 $\Rightarrow F + T$
 $\Rightarrow D + T$
 $\Rightarrow 1 + T$
 $\Rightarrow 1 + F$
 $\Rightarrow 1 + D$
 $\Rightarrow 1 + 2$


Sentential form 내
 논터미널 심볼이 여러
 개(예: E 와 T)인 경우
 가장 좌측에 있는
 논터미널(예: E)부터
 규칙을 적용하기로 함
 → **좌단유도**라고 함
 (뒤에 다시 설명됨)

- 문법 G를 사용하여 $1 + 2 * 3$ 의 유도과정을 보이시오

S
 $\Rightarrow E$
 $\Rightarrow E + T$
 $\Rightarrow T + T$
 $\Rightarrow F + T$
 $\Rightarrow D + T$
 $\Rightarrow 1 + T$
 $\Rightarrow 1 + T * F$
 $\Rightarrow 1 + F * F$
 $\Rightarrow 1 + D * F$
 $\Rightarrow 1 + 2 * F$
 $\Rightarrow 1 + 2 * D$
 $\Rightarrow 1 + 2 * 3$

- 문법 G가 생성하는 문장의 예를 보이시오

$1 * (2 + 3) * 4$



이 식을 유도하려면
 $1 * (2 + 3) * 4$

좌측 우선 유도 방향 $D * (E + T) * D$

S
 $\Rightarrow E$
 $\Rightarrow T$
 $\Rightarrow T * F$
 $\Rightarrow T * T * F$
 $\Rightarrow T * F * F$
 $\Rightarrow F * F * F$
 $\Rightarrow D * F * F$
 $\Rightarrow 1 * F * F$
 $\Rightarrow 1 * (E) * F$
 $\Rightarrow 1 * (2 + 3) * 4$

이제 좌측을 풀려면
 한 단계 뒤로 향하는 식이다

유도 연습 2

구문분석과 의미 분석을 한 번에 하기 쉬운 방법은
시간 복잡도 $O(n)$ 의 시간 이 걸림
같이 진행 하면 비선형 알고리즘 이용해야 함

V는 비어있는 집합
T는 비어있는 집합의 집합
Start는 시작
생성 규칙을 P라고 한다

문법 $G = (V, T, S, P)$
 $V = \{Program, Stmts, S, ID, NUM\}$
 $T = \{ \{, \}, =, ;, , print, (,), a, b, c, 0, 1, 2, 3, 4, 5, 6, 7, 8, 9 \}$

문법 G의 생성규칙집합 P

$Program \rightarrow \{ Stmts \}$
 $Stmts \rightarrow S Stmts \mid \epsilon$
 $S \rightarrow ID = NUM ;$
 $S \rightarrow print (ID);$
 $ID \rightarrow a \mid b \mid c$
 $NUM \rightarrow 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$

- 문법 G를 사용하여 $\{a = 7;\}$ 의 유도과정을 보이시오

$Program$
 $\Rightarrow \{ Stmts \}$
 $\Rightarrow \{ S Stmts \}$
 $\Rightarrow \{ ID = NUM ; Stmts \}$
 $\Rightarrow \{ a = NUM ; Stmts \}$
 $\Rightarrow \{ a = 7 ; Stmts \}$
 $\Rightarrow \{ a = 7 ; \}$

$Program \Rightarrow \{ Stmts \}$
 $\Rightarrow \{ S Stmts \}$
 $\Rightarrow \{ ID = NUM ; Stmts \}$
 $\Rightarrow \{ a = NUM ; Stmts \}$
 $\Rightarrow \{$

- 문법 G를 사용하여 $\{b = 5; print(b); \}$ 의 유도과정을 보이시오

$Program$
 $\Rightarrow \{ Stmts \}$
 $\Rightarrow \{ S Stmts \}$
 $\Rightarrow \{ ID = NUM ; Stmts \}$
 $\Rightarrow \{ b = NUM ; Stmts \}$
 $\Rightarrow \{ b = 5 ; Stmts \}$
 $\Rightarrow \{ b = 5 ; S Stmts \}$
 $\Rightarrow \{ b = 5 ; print(ID) ; Stmts \}$
 $\Rightarrow \{ b = 5 ; print(b) ; Stmts \}$
 $\Rightarrow \{ b = 5 ; print(b) ; \}$

- 문법 G로부터 $\{c = 3\}$ 을 유도해 보시오
- 문법 G로부터 $\{c = 3; print(b); \}$ 를 유도해 보고, 프로그래밍언어 관점에서 어떤 문제가 있는지 생각해 보시오.

Lexeme
 $Token \rightarrow ((ID, b)$
 (ID, b)
 $(ASSIGN, =)$
 $(NUM, 5)$
 \vdots

유도 연습 3

$Program \rightarrow \{ Stmts \}$
 $Stmts \rightarrow S Stmts \mid \epsilon$
 $S \rightarrow ID = E ;$
 $S \rightarrow print (ID);$
 $S \rightarrow while (E) S$
 $S \rightarrow \{ Stmts \}$
 $E \rightarrow E ! = E \mid T$
 $T \rightarrow T + F \mid T - F \mid F$
 $F \rightarrow ID \mid NUM$
 $ID \rightarrow a \mid b \mid c$
 $NUM \rightarrow 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$

구문문맥

어휘문맥

- 위 문법을 사용하여 아래 문장의 유도과정을 보이시오

$\{ a = 1; while(a \neq 9) \{ print(a); a = a + 1; \} \}$

Program $\Rightarrow \{ Stmts \}$

$\Rightarrow \{ S Stmts \}$
 $\Rightarrow \{ ID = E; Stmts \}$
 $\Rightarrow \{ ID = T; Stmts \}$
 $\Rightarrow \{ ID = NUM; Stmts \}$
 $\Rightarrow \{ ID = NUM; S Stmts \}$
 $\Rightarrow \{ ID = NUM; while(E) S Stmts \}$
 $\Rightarrow \{ ID = NUM; while(E) \{ Stmts \} Stmts \}$

Program

$\Rightarrow \{ Stmts \}$
 $\Rightarrow \{ S Stmts \}$
 $\Rightarrow \{ ID = E; Stmts \}$
 $\Rightarrow \{ a = E; Stmts \}$
 $\Rightarrow \{ a = T; Stmts \}$
 $\Rightarrow \{ a = F; Stmts \}$
 $\Rightarrow \{ a = NUM; Stmts \}$
 $\Rightarrow \{ a = 1; Stmts \}$
 $\Rightarrow \{ a = 1; S Stmts \}$
 $\Rightarrow \{ a = 1; while(E) S Stmts \}$
 $\Rightarrow \{ a = 1; while(E \neq E) S Stmts \}$
 $\Rightarrow \{ a = 1; while(T \neq E) S Stmts \}$
 $\Rightarrow \{ a = 1; while(F \neq E) S Stmts \}$
 $\Rightarrow \{ a = 1; while(ID \neq E) S Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq E) S Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq T) S Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq F) S Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq NUM) S Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) S Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ S Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(ID); Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); S Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); ID = E; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = E; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = T; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = T + F; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = F + F; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = ID + F; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = a + F; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = a + NUM; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = a + 1; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = a + 1; \} Stmts \}$
 $\Rightarrow \{ a = 1; while(a \neq 9) \{ print(a); a = a + 1; \} \}$

예제

while(b != 3)
print(b)

어휘분석
↓
구문분석
↓
의미분석
↓
중간 표현 생성
↓
코드 생성

토큰 4개를 입력 받음
파싱 해서 AST를 생성한다.

파싱을 함 (유도 트리를 찾는 것)
⇒ 유도 트리 (파스트리)
파스트리 → AST

유도 과정

Program ⇒ { S stmts }

⇒ { S stmts }

⇒ { while(E) S stmts }

⇒ { while(E != E) S stmts }


⇒ { while(T != E) S stmts }

⇒ { while(b != 3) S stmts }

⇒ { while(b != 3) print(ID); stmts }


⇒ { while(b != 3) print(b); }

Parse Tree 구현



다음 강 확인 #

9/ Parse Tree를 AST로 바꾸기



※ 시험

해당 코드에서 리터럴을 모두 찾으시오

해당 코드에서 연산자를 모두 찾으시오

유도 연습 4

```

Program → Subpgms
Subpgms → Subpgms Subpgm | Subpgm
Subpgm → def Id ( ) { Stmts } | def Id ( Params ) { Stmts }
Params → Params , Id | Id
Stmts → Stmts S | ε
S → Id = E ; | read ( Id ) ; | write ( E ) ; | return ; | return E ; | E ; | ;
S → if ( E ) S | if ( E ) S else S
S → while ( E ) S
S → { Stmts }
E → F | E < F | E <= F | E > F | E >= F
F → G | F + G | F - G
G → H | G * H | G / H | G % H
H → Id | Num | Str | ( E ) | Id ( ) | Id ( Args )
Args → Args , E | E
  
```

- 위 문법을 사용하여 아래 문장의 유도과정을 보이시오

```

def fact(n){
  if(n<=1) return 1;
  return n*fact(n-1);
}


def main() {
  read(n);
  write("Result=");
  write(fact(n));
}
  
```

추후 어휘분석기를 구현할 때 알아야 할 것

(ID, fact) — 마지막 'fact' ID가 여러 개 있을 수 있다.
(LPAREN, ()) — 그러나 LPAREN의 개수와 RPAR의 개수가 같아서 ((,))로 조합 가능하다.

```

Program
⇒ Subpgms
⇒ Subpgms Subpgm
⇒ Subpgm Subpgm
⇒ def Id ( Params ) { Stmts } Subpgm
⇒ def Id ( Id ) { Stmts } Subpgm
⇒ def Id ( Id ) { Stmts S } Subpgm
⇒ def Id ( Id ) { S S } Subpgm
⇒ def Id ( Id ) { if ( E ) S S } Subpgm
⇒ def Id ( Id ) { if ( E <= F ) S S } Subpgm
⇒ def Id ( Id ) { if ( F <= F ) S S } Subpgm
⇒ def Id ( Id ) { if ( G <= F ) S S } Subpgm
⇒ def Id ( Id ) { if ( H <= F ) S S } Subpgm
⇒ def Id ( Id ) { if ( Id <= F ) S S } Subpgm
⇒ def Id ( Id ) { if ( Id <= G ) S S } Subpgm
⇒ def Id ( Id ) { if ( Id <= H ) S S } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) S S } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return E ; S } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return F ; S } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return G ; S } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return H ; S } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; S } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return E ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return F ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return G ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return G * H ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return H * H ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * H ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( Args ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( E ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( F ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( F - G ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( G - G ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( H - G ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( Id - G ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( Id - H ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( Id - Num ) ; } Subpgm
⇒ def Id ( Id ) { if ( Id <= Num ) return Num ; return Id * Id ( Id - Num ) ; } def Id ( ) { Stmts }
...
  
```



해당 코드를 유도 할 때
유도가 되면 오류 X
유도가 안된다면 오류 발생

Leftmost derivation, rightmost derivation

Leftmost derivation (좌단유도)

- 유도의 각 단계에서 sentential form 내 leftmost non-terminal symbol(가장 왼쪽 논터미털 심볼)을 교체하여 얻어진 derivation을 leftmost derivation이라 함

Rightmost derivation (우단유도)

- 유도의 각 단계에서 sentential form 내 rightmost non-terminal symbol(가장 오른쪽 논터미털 심볼)을 교체하여 얻어진 derivation을 rightmost derivation이라 함

$$\begin{aligned}
 E &\rightarrow E + T \mid E - T \mid T \\
 T &\rightarrow T * F \mid T / F \mid F \\
 F &\rightarrow (E) \mid id
 \end{aligned}$$

$id + (id * id)$ 의 leftmost derivation	$id + (id * id)$ 의 rightmost derivation
E $\Rightarrow E + T$ $\Rightarrow T + T$ $\Rightarrow F + T$ $\Rightarrow id + T$ $\Rightarrow id + F$ $\Rightarrow id + (E)$ $\Rightarrow id + (T)$ $\Rightarrow id + (T * F)$ $\Rightarrow id + (F * F)$ $\Rightarrow id + (id * F)$ $\Rightarrow id + (id * id)$	E $\Rightarrow E + T$ $\Rightarrow E + F$ $\Rightarrow E + (E)$ $\Rightarrow E + (T)$ $\Rightarrow E + (T * F)$ $\Rightarrow E + (T * id)$ $\Rightarrow E + (F * id)$ $\Rightarrow E + (id * id)$ $\Rightarrow T + (id * id)$ $\Rightarrow F + (id * id)$ $\Rightarrow id + (id * id)$

유도의 트리 표현 1


문법 $G = (V, T, S, P)$
 $V = \{S, E, T, F, D\},$
 $T = \{+, *, (,), 0, 1, 2, 3, 4, 5, 6, 7, 8, 9\}$

문법 G의 생성규칙집합 P

$S \rightarrow E$
 $E \rightarrow E + T$
 $E \rightarrow T$
 $T \rightarrow T * F$
 $T \rightarrow F$
 $F \rightarrow (E)$
 $F \rightarrow D$
 $D \rightarrow 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9$

- 문법 G로부터 $1 + 2 * 3$ 의 유도 과정과 유도 트리

S
 $\Rightarrow E$
 $\Rightarrow E + T$
 $\Rightarrow T + T$
 $\Rightarrow F + T$
 $\Rightarrow D + T$
 $\Rightarrow 1 + T$
 $\Rightarrow 1 + T * F$
 $\Rightarrow 1 + F * F$
 $\Rightarrow 1 + D * F$
 $\Rightarrow 1 + 2 * F$
 $\Rightarrow 1 + 2 * D$
 $\Rightarrow 1 + 2 * 3$



- 유도트리로부터 문장의 구문 구조 파악 가능
- $(1 + 2) * 3$ 가 아니라 $1 + (2 * 3)$ 임

유도의 트리 표현 2

$Program \rightarrow \{ Stmts \}$

$Stmts \rightarrow S Stmts \mid \epsilon$

$S \rightarrow ID = E ;$

$S \rightarrow print (ID) ;$

$S \rightarrow while (E) S$

$S \rightarrow \{ Stmts \}$

$E \rightarrow E ! = E \mid T$


$T \rightarrow T + F \mid T - F \mid F$

$F \rightarrow ID \mid NUM$

$ID \rightarrow a \mid b \mid c$

$NUM \rightarrow 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$

Program
 $\Rightarrow \{ Stmts \}$
 $\Rightarrow \{ S Stmts \}$
 $\Rightarrow \{ ID = E ; Stmts \}$
 $\Rightarrow \{ a = E ; Stmts \}$
 $\Rightarrow \{ a = T ; Stmts \}$
 $\Rightarrow \{ a = F ; Stmts \}$
 $\Rightarrow \{ a = NUM ; Stmts \}$
 $\Rightarrow \{ a = 1 ; Stmts \}$
 $\Rightarrow \{ a = 1 ; S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (E) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (E ! = E) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (T ! = E) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (F ! = E) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (ID ! = E) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = E) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = T) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = F) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = NUM) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) S Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ S Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (ID) ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; S Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; ID = E ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = E ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = T ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = T + F ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = F + F ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = ID + F ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = a + F ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = a + NUM ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = a + 1 ; Stmts \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = a + 1 ; \} Stmts \}$
 $\Rightarrow \{ a = 1 ; while (a ! = 9) \{ print (a) ; a = a + 1 ; \}$



- 문장 $\{ a = 1 ; while (a ! = 9) \{ print (a) ; a = a + 1 ; \} \}$ 의 유도 트리
- 유도트리로부터 문장의 구문 구조 파악 가능

Parsing

Parsing(파싱)

- 문법 G 와 스트링 w 가 주어질 때, G 로부터 w 가 유도될 때 사용된 생성규칙들의 나열을 찾는 과정을 파싱(parsing)이라 함

문법 $S \rightarrow SS \mid aSb \mid bSa \mid \lambda$ 에 대해 스트링 $w = aabb$ 를 파싱

- $S \Rightarrow aSb \Rightarrow aaSbb \Rightarrow aabb$

- 소스프로그램이 언어의 문법에 맞게 작성된 것인지 확인하는 것으로 파싱 방법에는 top-down parsing(하향식 파싱)과 bottom-up parsing(상향식 파싱)이 있음

Top-down parsing(하향식 파싱)

- 시작 심볼에서 시작하여 생성 규칙을 적용하여 입력 스트링을 유도하는 방법
- 시작 심볼에서부터 하향식 트리 확장을 통해 parse tree를 찾는 방법
- 입력 스트링의 좌단유도를 찾는 과정으로 볼 수 있음
- Recursive descent parsing(재귀하강파싱)과 LL parsing(LL 파싱) 방법이 있음
- GCC, V8 등 recursive descent parser로 구현됨

Bottom-up parsing(상향식 파싱)

- 입력 스트링에서 시작하여 생성 규칙을 반대 방향으로 적용하여 시작 심볼을 얻어내는 방법
- Shift-reduce parsing 방법이 있음

Parse tree

Parse tree (파스 트리, derivation tree, 유도 트리, concrete syntax tree)


- Derivation을 ordered rooted tree로 보인 것

$$E \rightarrow E + T \mid E - T \mid T$$

$$T \rightarrow T * F \mid T / F \mid F$$

$$F \rightarrow (E) \mid id$$

$id + (id * id)$ 의 leftmost derivation	$id + (id * id)$ 의 rightmost derivation
E	E
$\Rightarrow E + T$	$\Rightarrow E + T$
$\Rightarrow T + T$	$\Rightarrow E + F$
$\Rightarrow F + T$	$\Rightarrow E + (E)$
$\Rightarrow id + T$	$\Rightarrow E + (T)$
$\Rightarrow id + F$	$\Rightarrow E + (T * F)$
$\Rightarrow id + (E)$	$\Rightarrow E + (T * id)$
$\Rightarrow id + (T)$	$\Rightarrow E + (F * id)$
$\Rightarrow id + (T * F)$	$\Rightarrow E + (id * id)$
$\Rightarrow id + (F * F)$	$\Rightarrow T + (id * id)$
$\Rightarrow id + (id * F)$	$\Rightarrow F + (id * id)$
$\Rightarrow id + (id * id)$	$\Rightarrow id + (id * id)$



- $id + (id * id)$ 에 대한 leftmost derivation의 파스 트리와 rightmost derivation의 파스 트리가 동일함


Abstract syntax tree (syntax tree)

Abstract syntax tree(AST, syntax tree, 구문트리, 추상구문트리)

- (문법으로부터의 유도를 표현하는 parse tree와 달리) 프로그램의 필수 구조를 트리로 표현한 것 → parse tree와 달리 AST에서는 파싱 이후 단계에서 불필요한 요소들(예: 괄호, 세미콜론 등)은 표현되지 않음 → abstract syntax tree와의 구별을 강조하기 위해 parse tree를 concrete syntax tree라고도 부름
- 파싱이 수행되는 동안 AST 노드(node)들이 만들어지며 노드에 속성(attribute)값들이 추가됨
- AST의 노드는 하나의 programming construct(프로그램 구성 요소)에 대응되며 construct의 의미를 구성하는 노드들을 자식 노드로 갖는다

No	Production	Semantic rule
1	$E \rightarrow E_1 + T$	$E.node = new Node("+", E_1.node, T.node)$
2	$E \rightarrow E_1 - T$	$E.node = new Node("-", E_1.node, T.node)$
3	$E \rightarrow T$	$E.node = T.node$
4	$T \rightarrow (E)$	$T.node = E.node$
5	$T \rightarrow id$	$T.node = new Leaf("id", id.entry)$
6	$T \rightarrow num$	$T.node = new Leaf("num", num.val)$

- $new Leaf(op, val) \rightarrow$ 2개 필드를 갖는 단말 노드 생성. op 는 노드 레이블. val 은 노드의 lexical value
- $new Node(op, c_1, c_2, \dots, c_k) \rightarrow$ $k+1$ 개 필드를 갖는 내부 노드 생성. op 는 노드 레이블. c_i 는 i 번째 자식 노드
- $id.entry$ points to the symbol table




Parse tree를 후위순회(postorder traversal)하면서
AST 생성 시, 생성 규칙 적용 순서 → 5, 6, 2, 5, 1


Parse tree vs. syntax tree

$$E \rightarrow E + T \mid E - T \mid T$$
$$T \rightarrow T * F \mid T / F \mid F$$
$$F \rightarrow (E) \mid id$$

id + (*id* * *id*)



parse tree




AST


Parse tree vs. syntax tree

$Program \rightarrow \text{main} () \{ Stmts \}$
 $Stmts \rightarrow Stmt Stmts \mid \epsilon$
 $Stmt \rightarrow ID = Expr ;$
 $Stmt \rightarrow \text{print } Expr ;$
 $Expr \rightarrow Expr + Term \mid Term$
 $Term \rightarrow Term * Factor \mid Factor$
 $Factor \rightarrow (Expr) \mid NUM \mid ID$

ID	[a-z]
NUM	[0-9]+(\\. [0-9]*)



AST



```

main() {
  x=1;
  y=x+20*(3-4);
  print y;
}
  
```

parse tree

Ambiguous grammar (1/2)

Ambiguous grammar

- 하나의 문장에 대해 2개 이상의 서로 다른 파스 트리를 생성하는 문법을 ambiguous grammar라고 한다

$$E \rightarrow E + E \mid E - E \mid E * E \mid E / E \mid (E) \mid 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$$

3 + 4 * 5의 leftmost derivation, parse tree, 의미

E $\Rightarrow E + E$ $\Rightarrow 3 + E$ $\Rightarrow 3 + E * E$ $\Rightarrow 3 + 4 * E$ $\Rightarrow 3 + 4 * 5$	E $\Rightarrow E * E$ $\Rightarrow E + E * E$ $\Rightarrow 3 + E * E$ $\Rightarrow 3 + 4 * E$ $\Rightarrow 3 + 4 * 5$
<pre> graph TD E1[E] --- E2[E] E1 --- P1[+] E2 --- 3[3] E2 --- E3[E] E3 --- E4[E] E3 --- M1[*] E4 --- 4[4] E4 --- 5[5] </pre>	<pre> graph TD E1[E] --- E2[E] E1 --- M2[*] E2 --- E3[E] E2 --- P2[+] E3 --- 3[3] E3 --- E4[E] E4 --- 4[4] E4 --- 5[5] </pre>
의미 $\rightarrow 3 + (4 * 5) = 23$	의미 $\rightarrow (3 + 4) * 5 = 35$

- 3 + 4 * 5라는 하나의 문장에 대해 서로 다른 2개의 파스 트리가 생성되므로 **ambiguous** 문법임
- Ambiguous 문법의 경우 하나의 문장에 대해 서로 다른 해석을 갖는 서로 다른 파스 트리들이 생성 가능
- Ambiguous 문법에 대한 구문 분석기는 파스 트리를 결정적으로 구성하기 어려움 \rightarrow 2가지 해결 방법
- 해결 방법 ① ambiguous 문법을 unambiguous 문법으로 변환하여 사용. 그러나 모든 ambiguous 문법이 unambiguous 문법으로 변환 가능한 것은 아님
- 해결 방법 ② ambiguous 문법으로 얻어진 구문분석기의 파싱표 내 충돌을 제거

Ambiguous grammar (2/2)

Ambiguous grammar

- 하나의 문장에 대해 2개 이상의 서로 다른 파스 트리를 생성하는 문법을 ambiguous grammar라고 한다

$S \rightarrow IF (B) S$
 $S \rightarrow IF (B) S ELSE S$
 $S \rightarrow id = B ;$
 $B \rightarrow E > E \mid E$
 $E \rightarrow id \mid num$



$IF (age > 60)$
 $IF (point > 100)$
 $rate = 20;$
 $ELSE rate = 10;$

$IF (age > 60)$
 $IF (point > 100)$
 $rate = 20;$
 $ELSE rate = 10;$

아래 문장에 대해 2개 서로 다른 파스 트리가 생성되므로 위 문법은 ambiguous 문법임


if (age>60) if(point>100) rate=20; else rate=10;

이 문법은 코드는 같지만 괄호의 경우
의미로 해석될 수 있다.



Ambiguous 문법 \rightarrow unambiguous 문법


$E \rightarrow E + E$
 $E \rightarrow E * E$
 $E \rightarrow (E) \mid id \mid num$



3 + 4 * 5의 parse tree들

- 왼쪽 parse tree는 $3 + (4 * 5)$ 를 의미
- 오른쪽 parse tree는 $(3 + 4) * 5$ 를 의미

$E \rightarrow E + T \mid T$
 $T \rightarrow T * F \mid F$
 $F \rightarrow (E) \mid id \mid num$



- 하나의 parse tree가 얻어짐


Ambiguous 문법 → unambiguous 문법

Ambiguous grammar

$E \rightarrow E + E \mid E - E \mid E * E \mid E / E$
 $E \rightarrow (E)$
 $E \rightarrow 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$


3 - 4 - 5의 leftmost derivation, parse tree, 의미

E
 $\Rightarrow E - E$
 $\Rightarrow E - E - E$
 $\Rightarrow 3 - E - E$
 $\Rightarrow 3 - 4 - E$
 $\Rightarrow 3 - 4 - 5$



$(3 - 4) - 5 = -6$

E
 $\Rightarrow E - E$
 $\Rightarrow 3 - E$
 $\Rightarrow 3 - E - E$
 $\Rightarrow 3 - 4 - E$
 $\Rightarrow 3 - 4 - 5$




$3 - (4 - 5) = 4$


Unambiguous grammar

$E \rightarrow E + T \mid E - T \mid T$
 $T \rightarrow T * F \mid T / F \mid F$
 $F \rightarrow (E) \mid 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$

문장 3 + 4 * 5에 대해
유일한 파스트리 생성됨



문장 3 - 4 - 5에 대해
유일한 파스트리 생성됨




문법 작성: precedence, associativity 반영

우선순위 (precedence)	연산자 (operator)	결합규칙(associativity)
1	*, /	Left to right
2	+, -	Left to right

Operator precedence, associativity


- *, /가 +, -보다 우선순위가 높음
- *, /는 우선순위 동일하며, left associative
- +, -는 우선순위 동일하며, left associative

$$\begin{aligned}
 E &\rightarrow E + T \mid E - T \mid T \\
 T &\rightarrow T * F \mid T / F \mid F \\
 F &\rightarrow (E) \mid id \mid num
 \end{aligned}$$


$$1 + 2 + 3 + 4 * 5 / 6 * 7 = (((1 + 2) + 3) + (((4 * 5) / 6) * 7))$$

Operator precedence, associativity

- **가 +보다 우선순위가 높음
- **는 right associative
- +는 left associative

$$\begin{aligned}
 E &\rightarrow E + T \mid T \\
 T &\rightarrow F \mid F ** T \\
 F &\rightarrow (E) \mid id \mid num
 \end{aligned}$$


$$1 + 2 + 3 + 4 ** 2 ** 3 = (((1 + 2) + 3) + (4^{(2^3)}))$$

문법에 연산자 우선순위, 결합법칙 반영

연습 1

- 연산자 $+, -, *, /$ 에 좌측 결합성 적용
- 괄호 내 식을 가장 먼저 계산
- 연산자 $*, /$ 의 우선순위가 연산자 $+, -$ 보다 높음

$$E \rightarrow E + T \mid E - T \mid T$$

$$T \rightarrow T * F \mid T / F \mid F$$

$$F \rightarrow (E) \mid 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$$

연습 2

- 연산자 $+, -, *, /$ 에 좌측 결합성 적용
- 연산자 $^$ 에 우측 결합성 적용
- 괄호 내 식을 가장 먼저 계산
- 단항연산자 $-$ 의 우선순위가 거듭제곱연산자 $^$ 보다 높음
- 연산자 $^$ 의 우선순위가 연산자 $*, /$ 보다 높음
- 연산자 $*, /$ 의 우선순위가 연산자 $+, -$ 보다 높음

$$E \rightarrow E + T \mid E - T \mid T$$

$$T \rightarrow T * F \mid T / F \mid F$$

$$F \rightarrow P \mid P^F$$

$$P \rightarrow -P \mid H$$

$$H \rightarrow (E) \mid 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$$

연습 3

- 연산자 $*, -$ 에 좌측 결합성 적용
- 연산자 $+, /$ 에 우측 결합성 적용
- 괄호 내 식을 가장 먼저 계산
- 연산자 $+, /$ 의 우선순위가 연산자 $*, -$ 보다 높음

$$E \rightarrow E * T \mid E - T \mid T$$

$$T \rightarrow F \mid F + T \mid F / T$$

$$F \rightarrow (E) \mid 0 \mid 1 \mid 2 \mid 3 \mid 4 \mid 5 \mid 6 \mid 7 \mid 8 \mid 9$$

※ 상황 ※
좌결합성을 가진 T 방지해주어
우결합성을 거둬주도록
작성해라!

※ 상황 ※
좌결합성을 가진 T 항기어에서
우결합성을 가지도록
작성해야 함!

C 언어 연산자 우선순위, 결합규칙

우선순위 (precedence)	연산자 (operator)	설명	결합규칙(associativity)
<div> <div>높음</div> <div>↑</div> <div>↓</div> <div>낮음</div> </div>	++ --	후위 증가/감소(postfix increment/decrement)	Left to Right
	++ -- + - !	전위 증가/감소(prefix increment/decrement) Unary plus/minus Logical NOT	Right to Left
	* / %	Multiplication, division, modulo	Left to Right
	+ -	Addition, subtraction	Left to Right
	< <= > >=	LT, LE, GT, GE	Left to Right
	== !=	EQ, NE	Left to Right
	&&	Logical AND	Left to Right
		Logical OR	Left to Right
	=	Assignment	Right to Left

대입 연산자는 우결합성을 가짐

좌결합성과 우결합성

$$3 - 4 - 5 = -6$$

$$3 - (4 - 5) = 4$$

$$(3 - 4) - 5 = -6$$

좌결합성


$E \rightarrow E - T \mid T$

$T \rightarrow T * F \mid F$


우결합성

$E \rightarrow T - E \mid T$

$T \rightarrow F + T \mid F$



문법 작성: left factoring



Recursive descent parsing

다음 token
식별자 저장


$P \rightarrow \{ S \}$
 $S \rightarrow ID = E ;$
 $E \rightarrow ID \mid NUM$

```
int token, ID=258, NUM=259;
void match(int t){
    if(token!=t) { printf("error\n"); exit(1); }
    token=nextToken();
}
void P() {
    match('{');
    S();
    match('}');
}
void S() {
    match(ID); match('='); E(); match(';');
}
void E() {
    if (token==ID) match(ID);
    else match(NUM);
}
int main(void){
    token=nextToken();
    P();
    printf("parsing success\n");
}
```

Recursive descent parsing(재귀하강파싱)

- 파싱 → 소스프로그램이 언어의 문법에 맞게 작성된 것인지 확인하는 것
- Recursive descent parser(재귀하강파서)는 언어 문법의 non-terminal symbol에 대응하는 (recursive) 프로시저들의 모음으로 구현 → 각 non-terminal에 대응하는 하나의 프로시저 작성(RHS 내 terminal은 다음 토큰과 일치(match)하도록 구현, RHS 내 non-terminal은 대응하는 프로시저 호출로 구현)


{ n = 1 ; }



```
P()
{
    S()
    match(ID)
    match('=')
    E()
    match(NUM)
    match(';')
}
```

재귀하강파서 구현 연습 1

$P \rightarrow \{ S \}$
 $S \rightarrow ID = E ;$
 $E \rightarrow ID \mid NUM$



컴파일 및 실행

flex lex.l

gcc -o lexer.exe lex.yy.c -LC:\WGNUWin32\lib -lfl

lexer < input1.txt

parsing success

lexer < input2.txt

error: invalid token

lexer < input3.txt

error: eof not found

input1.txt

```
{ age = 23; }
```

input2.txt

```
{ age = 23 }
```

input3.txt

```
{ age = 23; } { v = 1; }
```

```

%{
#include <stdio.h>
int token, ID=258, NUM=259;
}%

%%
[0-9]+      { return NUM; }
[a-zA-Z][a-zA-Z0-9]* { return ID; }
[{}=;]      { return yytext[0]; }
[ \t\n]     { }
.           { return yytext[0]; }
%%
    
```

```

void P(); void S(); void E();
int nextToken(){ return yylex(); }
void match(int t){
    if(token!=t) { printf("error: invalid token\n"); exit(1); }
    token=nextToken();
}

void P() { match('{'); S(); match('}'); }
void S() { match(ID); match('='); E(); match(';'); }
void E() {
    if (token==ID) match(ID);
    else match(NUM);
}

int main(void){
    token=nextToken();
    P();
    if(0==nextToken()) printf("parsing success\n");
    else printf("error: eof not found\n");
}
    
```

lex.l

lex.l

flex


lex.yy.c

- **yylex()**
- 입력에서 읽은 토큰 반환
- 파일 끝(end-of-file) 도달 시 0 반환

재귀하강파서 구현 연습 2

$P \rightarrow \{ S \}$
 $S \rightarrow ID = E ;$
 $E \rightarrow ID$
 $E \rightarrow ID + E$

$P \rightarrow \{ S \}$
 $S \rightarrow ID = E ;$
 $E \rightarrow ID Etail$
 $Etail \rightarrow +ID Etail \mid \epsilon$



다음 토큰이 ; 이면
 $Etail \rightarrow \epsilon$ 적용

컴파일 및 실행

flex lex.l

gcc -o lexer.exe lex.yy.c -LC:\WGNUWin32\lib -lfl

lexer < input1.txt

parsing success

lexer < input2.txt

parsing success

input1.txt

```
{ sum = sum + n; }
```

input2.txt

```
{ a = b + c + d; }
```

```

%{
#include <stdio.h>
int token, ID=258;
%}

%%
[a-zA-Z][a-zA-Z0-9]* { return ID; }
[{}=;+] { return yytext[0]; }
[ \t\n] { }
. { return yytext[0]; }
%%

void P(); void S(); void E(); void Etail();
int nextToken(){ return yylex(); }
void match(int t){
    if(token!=t) { printf("error: invalid token\n"); exit(1); }
    token=nextToken();
}

void P() { match('{'); S(); match('}'); }
void S() { match(ID); match('='); E(); match(';'); }
void E() { match(ID); Etail(); }
void Etail() {
    if (token==';') return;
    match('+'); match(ID); Etail(); // 재귀호출
}

int main(void){
    token=nextToken();
    P();
    if(0==nextToken()) printf("parsing success\n");
    else printf("error: eof not found\n");
}
    
```


문법 작성: left recursion 제거

Left recursive grammar

- Left-recursion에는 direct left-recursion과 indirect left-recursion이 있음
- Left-recursion이 포함된 문법을 left recursive라고 하며 left-recursive 문법은 recursive descent parsing에서 무한 재귀호출을 발생시킴

direct
left-recursion 발생

$$\begin{aligned} E &\rightarrow E + T \mid T \\ T &\rightarrow T * F \mid F \\ F &\rightarrow (E) \mid id \mid num \end{aligned}$$

recursive descent parser

```
...  
void E() {  
    E(); // 무한 재귀호출  
    match('+');  
    T();  
}  
...
```

Left-recursion
제거


$$\begin{aligned} E &\rightarrow T E' \\ E' &\rightarrow + T E' \mid \varepsilon \\ T &\rightarrow F T' \\ T' &\rightarrow * F T' \mid \varepsilon \\ F &\rightarrow (E) \mid id \mid num \end{aligned}$$

- Left-recursion이 제거된 문법
- Top-down parsing에 사용 가능

Shift-reduce parsing

Shift-reduce parsing

- Shift-reduce parsing은 shift와 reduce 동작에 기반한 파싱 방법으로, 파스 트리를 상향식(bottom-up)으로 점증적으로(incrementally) 만드는 효과가 있다
- Shift(이동) → 스택의 톱에서 handle 미발견 시, 다음 입력 심볼을 스택으로 옮기는 것
- Reduce(감축) → 스택의 톱에서 발견된 handle을, handle이 RHS에 매치되는 규칙의 LHS 심볼로 대체하는 것
- Accept(수락) → 초기 공백 스택과 입력 문장에서 시작하여, shift와 reduce 동작을 거친 후, 입력을 소진하고 스택의 톱에 문법의 start symbol이 발견되면 입력 문장을 문법에 맞는 문장으로 받아들이는 것






Shift-reduce parsing

Bottom-up
parser의 종류 중 하나임

Shift-reduce parsing

- Shift와 reduce 동작의 연속으로 수행되는 파싱을 shift-reduce parsing이라 함
- Shift(이동) → 스택의 톱에서 handle 미발견 시, 다음 입력 심볼을 스택으로 옮기는 것
- Reduce(감축) → 스택의 톱에서 발견된 handle을, handle이 RHS에 매치되는 규칙의 LHS 심볼로 대체하는 것
- Accept(수락) → 초기 공백 스택과 입력 문장에서 시작하여, shift와 reduce 동작을 거친 후, 입력을 소진하고 스택의 톱에 문법의 start symbol이 발견되면 입력 문장을 문법에 맞는 문장으로 받아들이는 것

왼쪽에서부터 역으로 제약을 함.



$E \rightarrow E + T \mid T$
 $T \rightarrow T * F \mid F$
 $F \rightarrow (E) \mid id$

스택	입력	동작	규칙
\$	$id + id * id\$$	Shift id	
$\$id$	$+ id * id\$$	Reduce	$F \rightarrow id$
$\$F$	$+ id * id\$$	Reduce	$T \rightarrow F$
$\$T$	$+ id * id\$$	Reduce	$E \rightarrow T$
$\$E$	$+ id * id\$$	Shift $+$	
$\$E +$	$id * id\$$	Shift id	
$\$E + id$	$* id\$$	Reduce	$F \rightarrow id$
$\$E + F$	$* id\$$	Reduce	$T \rightarrow F$
$\$E + T$	$* id\$$	Shift $*$	
$\$E + T *$	$id\$$	Shift id	
$\$E + T * id$	$\$$	Reduce	$F \rightarrow id$
$\$E + T * F$	$\$$	Reduce	$T \rightarrow T * F$
$\$E + T$	$\$$	Reduce	$E \rightarrow E + T$
$\$E$	$\$$	Accept	


$id + id * id$ 의
 rightmost
 derivation:
 E
 $\Rightarrow E + T$
 $\Rightarrow E + T * F$
 $\Rightarrow E + T * id$
 $\Rightarrow E + F * id$
 $\Rightarrow E + id * id$
 $\Rightarrow T + id * id$
 $\Rightarrow F + id * id$
 $\Rightarrow id + id * id$

Shift-reduce conflict

```
if (age>60) if(point>100) rate=20; else rate=10;
```

IF (id > num) IF(id > num) id = num; ELSE id = num;

스택	입력	동작	규칙
\$	$IF(id > num) IF(id > num) id = num; ELSE id = num; \$$	Shift	
\$IF	$(id > num) IF(id > num) id = num; ELSE id = num; \$$	Shift	
\$IF($id > num) IF(id > num) id = num; ELSE id = num; \$$	Shift	
\$IF(id	$> num) IF(id > num) id = num; ELSE id = num; \$$	Reduce	$E \rightarrow id$
\$IF(E	$> num) IF(id > num) id = num; ELSE id = num; \$$	Shift	
\$IF(E >	$num) IF(id > num) id = num; ELSE id = num; \$$	Shift	
\$IF(E > num	$) IF(id > num) id = num; ELSE id = num; \$$	Reduce	$E \rightarrow num$
\$IF(E > E	$) IF(id > num) id = num; ELSE id = num; \$$	Reduce	$B \rightarrow E > E$
\$IF(B	$) IF(id > num) id = num; ELSE id = num; \$$	Shift	
...	
\$IF(B) IF(B) S	$ELSE id = num; \$$	Shift or Reduce	

$$\begin{aligned} S &\rightarrow IF (B) S \\ S &\rightarrow IF (B) S ELSE S \\ S &\rightarrow id = B ; \\ B &\rightarrow E > E \mid E \\ E &\rightarrow id \mid num \end{aligned}$$


Reduce-reduce conflict

References

- ✚ 박두순. (2016). (내공 있는 프로그래머로 길러주는)컴파일러의 이해. 한빛아카데미.
- ✚ 창병모. (2021). 프로그래밍 언어론 : 원리와 실제. 인피니티북스.
- ✚ Aho, A., Lam, M., Sethi, R., Ullman, J. (2006). Compilers: Principles, Techniques, and Tools (2nd Ed.). Addison Wesley.
- ✚ Nystrom, R. Crafting interpreter. <https://craftinginterpreters.com/>
- ✚ Sebesta, R. (2012). Concepts of Programming Languages (10th. ed.). Pearson.
- ✚ Thain, D. (2023). Introduction to Compilers and Language Design.
- ✚ Paxson, V. (1995). Flex, version 2.5.
- ✚ Donnelly, C., Stallman, R. (2008). Bison.
- ✚ LexAndYacc.pdf (epaperpress.com)
- ✚ Tom Niemann, LEX & YACC. <http://epaperpress.com/lexandyacc>